Option Explicit

Sub DocumentConnections()

    Dim DesApp As Designer.Application

    Dim Conn As Designer.Connection

    Dim CurrentApp As String

    Dim Rng As Excel.Range

    Dim RowNum As Long

    On Error GoTo ErrorHandler

    'some Excel housekeeping

    CurrentApp = Application.Caption

    Application.Cursor = xlWait

    Application.DisplayAlerts = False

    'start Designer and log in

    Set DesApp = New Designer.Application

    DesApp.Window.State = dsMinimized

    DesApp.Visible = True

    Application.StatusBar = "Logging in..."

    Call DesApp.LogonDialog

    DesApp.Visible = False

    'restore Excel as active application

    Call AppActivate(CurrentApp)

    Application.ScreenUpdating = True

    'document the connections

    Application.StatusBar = "Documenting connections..."

    Set Rng = Sheets("Connections").Cells

    RowNum = 1

    For Each Conn In DesApp.Connections

        RowNum = RowNum + 1

        Rng(RowNum, 1) = Conn.Name

        Rng(RowNum, 2) = Conn.DatabaseEngine

        Rng(RowNum, 3) = Conn.NetworkLayer

        Rng(RowNum, 4) = Conn.DatabaseSource

        Rng(RowNum, 5) = Conn.UserName

    Next Conn

CleanUp:

    On Error Resume Next

    Application.StatusBar = False

    Application.Cursor = xlDefault

    Set Conn = Nothing

    DesApp.Quit

    Set DesApp = Nothing

    Exit Sub

ErrorHandler:

    Call AppActivate(CurrentApp)

    MsgBox Err.Source & " - " & Err.Number & ":  " & Err.Description, _

        vbCritical, "Failure in DocumentConnections()"

    Resume CleanUp

End Sub

